Consider a [primitive permutation group](https://en.wikipedia.org/wiki/Primitive_permutation_group) $\Gamma\subseteq\mathrm{Sym}(N)$ on the $n$-element set $N=\{1,...,n\}$, that is, $\Gamma$ does not preserve any non-trivial partition of $N$.


Consider the linear representation of $\Gamma$ by permutation matrices on $\Bbb R^n$ in the obvious way. 
What can be said about the decomposition of this representation into irreducible factors?
I am especially interested in the multiplicities of the factors: is it known whether any two irreducible factors in such a decomposition are non-isomorphic?



Irreducible factors of primitive permutation group representation
An example in which there are two isomorphic irreducible modules in the decomposition is the group ${\rm PSL}(2,11)$ in its primitive permutation representation of degree $55$ coming from the action of $G$ on the cosets of a dihedral subgroup of order $12$. The permutation module over the real numbers decomposes into modules of dimensions $1,10,10,10,12,12$, where two of the $10$-dimensional constituents are isomorphic.


Here is a calculation in Magma that verifies this. I am doing this calculation over the complex field, where the decomposition is $1+5+5+10+10+12+12$, but note that the two $5$-dimensional constituents are contragredient, and they combine to make a $10$- real representation.



```
> G := PrimitiveGroup(55,1);
> ChiefFactors(G);
    G
    |  A(1, 11)                   = L(2, 11)
    1
> CT := CharacterTable(G);
> CT; 

Character Table of Group G
--------------------------

-------------------------------------------
Class |    1  2   3    4    5   6    7    8
Size  |    1 55 110  132  132 110   60   60
Order |    1  2   3    5    5   6   11   11
-------------------------------------------
p  =  2    1  1   3    5    4   3    8    7
p  =  3    1  2   1    5    4   2    7    8
p  =  5    1  2   3    1    1   6    7    8
p  = 11    1  2   3    4    5   6    1    1
-------------------------------------------
X.1   +    1  1   1    1    1   1    1    1
X.2   0    5  1  -1    0    0   1   Z2 Z2#2
X.3   0    5  1  -1    0    0   1 Z2#2   Z2
X.4   +   10 -2   1    0    0   1   -1   -1
X.5   +   10  2   1    0    0  -1   -1   -1
X.6   +   11 -1  -1    1    1  -1    0    0
X.7   +   12  0   0   Z1 Z1#2   0    1    1
X.8   +   12  0   0 Z1#2   Z1   0    1    1

Explanation of Character Value Symbols

# denotes algebraic conjugation, that is,
#k indicates replacing the root of unity w by w^k

Z1     = (CyclotomicField(5: Sparse := true)) ! [ RationalField() | 0, 0, 1, 1 ]    
Z2     = (CyclotomicField(11: Sparse := true)) ! [ RationalField() | 0, 1, 0, 1,
1, 1, 0, 0, 0, 1 ]

> K := CyclotomicField(55);
> M := PermutationModule(G,K);
> c := Character(M);
> Decomposition(CT,c);
[ 1, 1, 1, 0, 2, 0, 1, 1 ]

```

